
Changelog
    

This project adheres to Semantic Versioning.
v0.3.0 (2023-01-16)
Enhancements
	[Breaking / Janus.Policy] Remove :module field from %Janus.Policy{} struct.
	[Breaking / Janus.Policy] Change schema/action argument order in allow/4 and deny/4 to be more consistent with the rest of Janus. See example below.
	[Janus.Policy] Add Janus.Policy.build_policy/1 callback primarily to document its usage; an implementation was already injected into policy modules.
	[Janus.Policy] Add :actor field to %Janus.Policy{} struct.

Example allow/4 and deny/4
# Old argument order
policy
|> allow(:read, Thread, where: [...])
|> allow(:create, Thread, where: [...])
|> deny(:create, Thread, where: [...])

# New argument order
policy
|> allow(Thread, :read, where: [...])
|> allow(Thread, :create, where: [...])
|> deny(Thread, :create, where: [...])
v0.2.1 (2023-01-14)
Enhancements
	[Janus.Policy] Raise if something other than a schema is passed to allow/4 or deny/4.

v0.2.0 (2023-01-05)
Enhancements
	[Breaking / Janus.Policy] policy_for renamed to build_policy.
	[Breaking / Janus.Policy] before_policy_for removed.
	[Breaking / Janus.Authorization] filter_authorized renamed to scope.
	[Breaking / Janus.Authorization] authorize now returns {:error, :not_authorized} instead of :error.
	[Janus] use Janus accepts optional defaults for Janus.Authorization.authorize/4 options.
	[Janus.Policy] Pre-authorization hooks with Janus.Policy.attach_hook/4 and friends.
	[Janus.Authorization] authorize/4 adds :repo and :load_assocations options.
	[mix janus.gen.policy] Overhaul generated policy helpers to more easily replace Ecto.Repo callbacks.

v0.1.0 (2022-12-25)
First release.

Setup
Generator
$ mix janus.gen.policy
* creating lib/my_app/policy.ex

Generate a starting policy module for your application.
For more information, see the Generated Policy Modules guide.
Policy module
Your policy module is the interface used by the rest of your application.
This is usually the only place you should be referring to Janus directly.
lib/my_app/policy.ex
defmodule MyApp.Policy do
  use Janus

  @impl true
  def build_policy(policy, actor) do
    policy
    |> # authorization rules
  end
end
For more information on defining policies, see the Policy Definition Cheatsheet
Core concepts
Schemas
Schemas are modules that use Ecto.Schema.
Used when defining policies
policy
|> allow(Post, :edit, ...)
#        ^^^^
Used when scoping a query
MyApp.Policy.scope(Post, :edit, current_user)
#                  ^^^^
Used when checking for permissions
MyApp.Policy.any_authorized?(Post, :edit, current_user)
#                            ^^^^
Resources
Resources are loaded structs defined by one of your schemas.
Used when authorizing an action
MyApp.Policy.authorize(%Post{}, :edit, current_user)
#                      ^^^^^^^
Actors
Actors are the users of your application.
They can be a %User{} struct, but they don't have to be.
Actors are converted to a policy using build_policy/2, so an actor can be anything that you want to use to differentiate between types of user.
They can even be a simple atom like :normal_user or :admin_user.
In build_policy/2
def build_policy(policy, %User{}) do
  #                      ^^^^^^^
end
Used when calling any authorization function
MyApp.Policy.authorize(%Post{}, :edit, current_user)
#                                      ^^^^^^^^^^^^

MyApp.Policy.scope(Post, :edit, current_user)
#                               ^^^^^^^^^^^^
Actions
Actions are what actors do to resources in your application.
Janus doesn't care how you represent actions, but atoms usually do the trick.
Used when defining policies
policy
|> allow(Post, :edit, ...)
#              ^^^^^
Used when calling any authorization function
MyApp.Policy.authorize(%Post{}, :edit, current_user)
#                               ^^^^^
Can be any term except a list
policy
|> allow(Post, :edit, ...)
|> allow(Post, "edit", ...)
|> allow(Post, %Action{type: :edit}, ...)
# lists are special-cased to allow multiple
# actions to share conditions
|> allow(Post, [:read, :edit],...)
Defining rules
Overview
Authorization rules are attached to policies in the build_policy/2 callback.
	Janus.Policy.allow/4
	Janus.Policy.deny/4
	Janus.Policy.allows/1

Grant permission for all resources of schema
policy
|> allow(Post, :read)
|> allow(Post, :edit)
|> allow(Post, :archive)
|> allow(Comment, :read)
|> allow(Comment, :edit)
Using lists of actions
policy
|> allow(Post, [:read, :edit, :archive], Post)
|> allow(Comment, [:read, :edit], Comment)
Grant permission based on attributes
policy
|> allow(Post, :read, where: [archived: false])

# or define using :where_not
|> allow(Post, :read, where_not: [archived: true])

# or override a blanket permission using deny
|> allow(Post, :read)
|> deny(Post, :read, where: [archived: true])
Use deny to override a previous allow
policy
|> allow(Post, :read)
|> deny(Post, :read, where: [archived: true])
Grant permission if the user is associated with the resource
def build_policy(policy, %User{role: :member} = user) do
  policy
  |> allow(Comment, :edit, where: [user_id: user.id])
end
Grant permission based on association attributes
policy
|> allow(Comment, :edit, where: [user: [role: :member]])
Use allows to delegate permission to an association
policy
|> allow(Post, :read, where: [archived: false])
|> allow(Comment, :read, where: [post: allows(:read)])
Multiple allow combines as a logical-or
# This will always allow reading all posts
policy
|> allow(Post, :read)
|> allow(Post, :read, where: [archived: false]) # has no effect
Hooks
Overview
Hooks are attached to policies in the build_policy/2 callback.
They are called prior to authorize or scope and can modify the resource/query or halt authorization altogether.
	Janus.Policy.attach_hook/4
	Janus.Policy.attach_new_hook/4
	Janus.Policy.detach_hook/3

Run prior to authorizing any schema
policy
|> attach_hook(:my_hook, fn
  :authorize, resource, _action ->
    {:cont, resource}

  :scope, query, _action ->
    {:cont, query}
end)
Run prior to authorizing a specific schema
policy
|> attach_hook(:my_hook, Post, fn
  :authorize, resource, _action ->
    {:cont, resource}

  :scope, query, _action ->
    {:cont, query}
end)
Used to preload fields
policy
|> attach_hook(:preload_user, Post, fn
  :authorize, post, _action ->
    {:cont, Repo.preload(post, :user)}

  :scope, query, _action ->
    {:cont, query}
end)
Remove attached hooks with detach_hook/3
policy
|> attach_hook(:my_hook, &my_hook/3)
|> detach_hook(:my_hook)
policy
|> attach_hook(:my_hook, Post, &my_hook/3)
|> detach_hook(:my_hook, Post)
Attach hook if it's new using attach_new_hook/4
# second call has no effect
policy
|> attach_new_hook(:my_hook, &my_hook/3)
|> attach_new_hook(:my_hook, &my_other_hook/3)
# second call has no effect
policy
|> attach_new_hook(:my_hook, Post, &my_hook/3)
|> attach_new_hook(:my_hook, Post, &my_other_hook/3)
# attaches second hook because :my_hook not added for Post
policy
|> attach_new_hook(:my_hook, &my_hook/3)
|> attach_new_hook(:my_hook, Post, &my_other_hook/3)
Run a late check before each authorization call
@impl true
def build_policy(policy, user) do
  policy
  |> attach_hook(:ensure_unbanned, fn _, object, _ ->
    if Accounts.banned?(user.id) do
      :halt
    else
      {:cont, object}
    end
  end)
  |> ...
end
This can be useful if policies are being cached. If the call to Accounts.banned?(user.id) occurred in the callback body instead, the policy could not react to any change in account status after it was built.
Structuring your policies
Pattern-match to give different permissions to different actors
def build_policy(policy, %User{role: :member}) do
  # member permissions
end

def build_policy(policy, %User{role: :moderator}) do
  # moderator permissions
end
Delegate to context-specific policies
def build_policy(policy, actor) do
  policy
  |> CommunityForum.Policy.build_policy(actor)
  |> Storefront.Policy.build_policy(actor)
end
For larger applications with well-defined boundaries, a policy can be constructed by threading it through multiple build_policy calls.

Authorization
Using policies in a context module
Policies should most often be used in context modules, since they provide the interface to actions and resources that the rest of your application uses.
defmodule MyApp.MessageBoard do
  @moduledoc """
  Context module for the message board.
  """
  import Ecto.Query

  # imports authorize, any_authorized?, scope, etc.
  import MyApp.Policy

  alias MyApp.Repo

  # ...
end
Authorizing an action using authorize/4
def update_post(%Post{} = post, attrs \\ %{}, user_or_policy) do
  case authorize(post, :edit, user_or_policy) do
    {:ok, post} ->
      post
      |> Post.changeset(attrs)
      |> Repo.update()

    {:error, :not_authorized} ->
      {:error, :not_authorized}
  end
end
iex> update_post(authorized_post, %{}, some_user)
{:ok, %Post{}}

iex> update_post(unauthorized_post, %{}, some_user)
{:error, :not_authorized}
Fetching authorized resources
def authorized_posts(user_or_policy) do
  if any_authorized?(Post, :read, user_or_policy) do
    posts =
      Post
      |> scope(:read, user_or_policy)
      |> Repo.all()

    {:ok, posts}
  else
    {:error, :not_authorized}
  end
end
Use any_authorized?/3 to differentiate between a result that is empty because there are no resources that match the policy conditions and a result that is empty because the user isn't authorized to view any resources.
Preloading authorized associations
scope(Post, :read, user_or_policy,
  preload_authorized: :comments
)
scope(Post, :read, user_or_policy,
  preload_authorized: [comments: :user]
)
The :preload_authorized option can be passed to preload only those associated resources that are authorized for the given action.
Applying a query to preloads
latest_comment_query =
  from Comment,
    order_by: [desc: :inserted_at],
    limit: 1

scope(Post, :read, user_or_policy,
  preload_authorized: [comments: latest_comment_query]
)
A query can be applied to associated authorized resources.
It is scoped per-association, so it applies to comments of each post instead of the comments of all posts.
The above would return all :read-able posts preloaded with their latest :read-able comment.
You can still include nested preloads using a tuple:
scope(Post, :read, user_or_policy,
  preload_authorized: [comments: {latest_comment_query, [:user]}]
)
Caching a policy
Call build_policy/1 to get the policy for a user
iex> policy = MyApp.Policy.build_policy(current_user)
%Janus.Policy{...}
Pass a policy anywhere you'd pass in an actor
iex> MyApp.Policy.authorize(post, :read, policy)
{:ok, post}

iex> MyApp.Policy.scope(Post, :read, policy)
%Ecto.Query{}
Cache a policy in a Plug.Conn
Call in a plug
def assign_current_policy(conn) do
  %{assigns: %{current_user: user}} = conn

  conn
  |> assign(:current_policy, MyApp.Policy.build_policy(user))
end
Controller action
def index(conn, _params) do
  %{assigns: %{current_policy: policy}} = conn

  # Pass the policy to your context
  case MessageBoard.authorized_posts(policy) do
    {:ok, posts} ->
      ...

    {:error, :not_authorized} ->
      ...
  end
end

Work in progress
This guide is incomplete, though its contents may still be useful if you're getting started.

Janus defines a small but flexible API that can be used to create a higher-level authorization interface for your application.
To support common conventions, like the use of Phoenix-style context modules, Janus provides a policy generator that defines helpers that are especially useful in those circumstances.
The goal of this guide is to explain the usage of, and reasoning behind, those helpers.
Generating a policy module
The policy generator creates a policy module containing authorization functions that can be used by the rest of your application.
$ mix janus.gen.policy [--module Example.Policy] [--path example/path/policy.ex]

When run without arguments, it will generate a policy module called YourApp.Policy at lib/your_app/policy.ex (with YourApp replaced by your actual application namespace).
Overview of helpers
Most of the generated helpers are designed to wrap common Ecto.Repo CRUD operations in a way that facilitates refactoring existing unauthorized calls to authorized ones with minimal necessary changes.
The helpers don't cover everything, but they cover a large portion of use-cases and provide a pattern for wrapping additional operations if needed.
The following helpers are included in newly-generated policy modules:
	authorized_fetch_by - wraps Ecto.Repo.get_by/3, gets a resource using the given attributes and then authorizes for the given action/user

	authorized_fetch_all - wraps Ecto.Repo.all/2, fetch a list of resources that are authorized for the given action/user

	authorized_insert - wraps Ecto.Repo.insert/2, operates on a changeset, failing with a validation error if the inserted resource would not be authorized for the given action/user

	authorized_update - wraps Ecto.Repo.update/2, operates on a changeset, failing with a validation error if the updated resource would not be authorized for the given action/user either before or after applying changes

	authorized_delete - wraps Ecto.Repo.delete/2, deletes the given resource if it is authorized for the given action/user

	validate_authorized - changeset validation that ensures the resource being changed is authorized, adding a validation error otherwise.


We'll go over each of these and how they might be used in the sections that follow.
Note on generated code
Remember that generated code is your code!
It should be modified, replaced, and deleted as you see fit.
Instead of including these helpers in Janus, they are generated as a starting point for your application.
They enable the usage patterns described below, but if they aren't useful to you, toss them and use whatever you see fit!

Authorized operations
The goal is to transition from unauthorized to authorized operations as smoothly as possible.
authorized_* functions take the same arguments as their Ecto.Repo counterparts, except that they add additional keyword options related to authorization.
Let's look at some examples.
iex> Repo.get_by(Post, id: 12345)
%Post{}

iex> Policy.authorized_fetch_by(Post, [id: 12345], authorize: {:read, user})
{:ok, %Post{}}
# or
{:error, :not_authorized}
# or
{:error, :not_found}
All of the authorized_* functions take the :authorize keyword option, which should be either a tuple of {action, actor} as above, or false to skip authorization altogether.
An ArgumentError will be raised if :authorize is not present, but this can be changed if you'd prefer authorization to be opt-in instead of opt-out.
(Opt-out, the default, is the recommended approach, though it may require more refactoring work up-front.)
There's another obvious difference in the example above: authorized_fetch_by/3 returns an :ok or :error tuple instead of the resource or nil.
Returning a tuple allows us to differentiate between a lookup failure and an auth failure.
A similar approach is used in the next example:
iex> Repo.all(Post)
[%Post{}, ...]

iex> Policy.authorized_fetch_all(Post, authorize: {:read, user})
{:ok, [%Post{}, ...]}
# or
{:error, :not_authorized}
A tuple is also returned here to differentiate authorization failures and an empty result.
An Ecto query can also be passed as the first argument.
iex> query = from p in Post, where: p.inserted_at > ago(1, "month")

iex> Repo.all(query)
[%Post{}, ...]

iex> Policy.authorized_fetch_all(query, authorize: {:read, user})
{:ok, [%Post{}, ...]}
# or
{:error, :not_authorized}

Authorization superpowers for applications using Ecto.
Priorities:
	Single source of truth - The same rules that authorize loaded data
should be able to load authorized data.

	Authentication-agnostic - Janus should not care about how users
are modeled or authenticated.

	Minimal library footprint - Expose a small but flexible API that
can be used to create an optimal authorization interface for each
application.

	Escape hatches where necessary - Complex authorization rules and
use-cases should be representable when Janus neglects to provide a
short cut.


Janus is split into two primary components:
	Janus.Policy - functions and behaviour for defining policy
modules, which describe the allowed actors, actions, and resources
in your application. This is where you look if you're writing a
policy module.

	Janus.Authorization - functions and behaviour used by the rest
of your application to authorize and load resources. This is where
you look if you're using a policy module.


Janus defines a Mix task to generate the basic policy module that will
get you started:
$ mix janus.gen.policy

Installation
Janus can be installed by adding ex_janus to your deps in mix.exs:
defp deps do
  [
    {:ex_janus, "~> 0.3.0"}
  ]
end
Policies
Policy modules are created by invoking use Janus, which implements
both the Janus.Policy and Janus.Authorization behaviours:
defmodule Policy do
  use Janus

  @impl true
  def build_policy(policy, _actor) do
    policy
  end
end
When you invoke use Janus, default implementations are injected for
required callbacks, except for Janus.Policy.build_policy/2. This
callback is your foundation, as it returns the authorization policy
for an individual user of your application.
The policy above is not very useful (it doesn't allow anyone to do
anything) but that can be changed by using the Janus.Policy API to
define actions, resources, and conditions that make up your
authorization rules.
def build_policy(policy, %User{role: :moderator} = mod) do
  policy
  |> allow(Post, :read)
  |> allow(Post, [:edit, :archive, :unarchive], where: [user: [role: :member]])
  |> allow(Post, [:edit, :archive, :unarchive], where: [user_id: mod.id])
  |> deny(Post, :unarchive, where: [archived_by: [role: :admin]])
end
See the Janus.Policy documentation for more on defining policies.
Authorization
With our policy module defined, it can now be used to load and
authorize resources.
iex> Policy.authorize(some_post, :archive, moderator)
{:ok, some_post}

iex> Policy.authorize(post_archived_by_admin, :unarchive, moderator)
{:error, :not_authorized}

iex> Policy.scope(Post, :read, moderator)
%Ecto.Query{}

iex> Policy.scope(Post, :read, moderator) |> Repo.all()
[ ... posts the moderator can read ]

iex> Policy.any_authorized?(Post, :edit, moderator)
true # there are rules allowing moderators to edit posts

iex> Policy.any_authorized?(Post, :delete, moderator)
false # there are no rules that allow moderators to delete posts
These functions make up the Janus.Authorization behaviour, and their
definitions were injected by default when we invoked use Janus. This
is the "public API" that the rest of your application will use to
authorize resources.
See the Janus.Authorization documentation for more.
Integration with Ecto.Query
The primary assumption that Janus makes is that your resources are
backed by an Ecto.Schema. Using Ecto's schema reflection
capabilities, Janus is able to use the same policy to authorize a
single resource and to construct a composable Ecto query that is aware
of field types and associations.
# This query would result in the 5 latest posts that the current
# user is authorized to see, preloaded with the user # who made
# the post (but only if the current user is  allowed to see that
# user).

Post
|> Policy.scope(:read, current_user,
  preload_authorized: :user
)
|> order_by(desc: :inserted_at)
|> limit(5)
This integration with Ecto queries is main reason Janus exists.
Configuration
Some defaults can be configured by passing them as options when
invoking use Janus. Those are:
	:repo - Ecto.Repo used to load associations when required by
your authorization rules

	:load_associations - Load associations when required by your
authorization rules (requires :repo config option to be set or
to be passed explicitly at the call site), defaults to false


For example:
defmodule MyApp.Policy do
    use Janus,
      repo: MyApp.Repo,
      load_associations: true

    # ...
end
These defaults will be referenced in the Janus.Authorization
documentation where they are used.
Why (not) Janus?
Janus was created to scratch an itch: the same rules that authorize
loaded data should be able to load authorized data. In concrete terms,
a rule that defines whether a user can edit a resource should also be
able to load all the resources that user can edit.
Loading data this way should be:
	efficient - loading everything and then filtering it in-memory
doesn't cut it;

	composable - it should be possible to add additional conditions
when loading data;

	ergonomic - authorization should slot-in naturally without major
rewrites.


Thankfully, integration with Ecto.Query solves for all of the above.
One only needs authorization rules that can be translated into a
query.
And thus, Janus was born.
Janus may be a good fit if...
	you're authorizing data backed by Ecto.Schema. Janus relies on
the reflection capabilities of schemas to produce correct queries,
cast values, navigate associations, etc.

	you share interfaces between users with different permissions.
Janus allows you to scope queries in a uniform way using the
current user (or lack of one), making shared interfaces a natural
default.

	you prefer to have the final say. Janus takes an approach similar
to Phoenix, generating code that supports certain conventions
while allowing you to override or redefine behavior to fit your
preferences.

	you prefer a functional API for defining rules. Authorization
policies are data; adding an authorization rule just transforms
that data. Policies can be built using the full extent and natural
composability of the Elixir language.


Janus may not be a good fit if...
	you're only authorizing actions that don't have an obvious
association to data backed by Ecto.Schema. For instance, a
:send_welcome_email action without some kind of Email schema.
Janus does, however, give you a natural place to define that sort
of API yourself (your policy module).

	you want an easy-to-read DSL for authorization rules. Janus
policies are "just code", so readability will depend on your own
style and structure. If you value readability/scannability very
highly, definitely check out LetMe,
which provides a great DSL and makes some different trade-offs
than Janus does.

	you want runtime introspection for your authorization rules, like
a list of all actions a user can perform. Janus does not currently
provide structured access to this information, but you might again
turn to LetMe, which provides
introspection capabilities.

Sets up a module to implement the Janus.Policy and
Janus.Authorization behaviours.
Using use Janus does the following:
	adds the Janus.Policy behaviour, imports functions used to
define the required callback Janus.Policy.build_policy/2, and
defines a build_policy/1 helper

	adds the Janus.Authorization behaviour and injects default
(overridable) implementations for all callbacks



  
  options

  
  Options


	:load_associations - Load associations when required by your
authorization rules (requires :repo config option to be set or
to be passed explicitly at the call site), defaults to false

	:repo - Ecto.Repo used to load associations when required by
your authorization rules


See "Configuration" section for details.

  
  example

  
  Example


defmodule MyApp.Policy do
  use Janus, repo: MyApp.Repo

  @impl true
  def build_policy(policy, _actor) do
    policy
    # |> allow(...)
  end
end

Authorize and load resources using policies.
Policy modules expose a minimal API that can be used to authorize and
load resources throughout the rest of your application.
	authorize/4 - authorize an individual, already-loaded resource

	scope/4 - construct an Ecto query for a schema that will
filter results to only those that are authorized

	any_authorized?/3 - checks whether the given actor/policy has
any access to the given schema for the given action


These functions will usually be called from your policy module
directly, since wrappers that accept either a policy or an actor are
injected when you invoke use Janus.  Documentation examples will
show usage from your policy module.
See individual function documentation for details.

          @callback authorize(
  Ecto.Schema.t(),
  Janus.action(),
  Janus.actor() | Janus.Policy.t(),

) ::
  {:ok, Ecto.Schema.t()} | {:error, :not_authorized}

          @spec any_authorized?(filterable(), Janus.action(), Janus.Policy.t()) :: boolean()


      


Checks whether any permissions are defined for the given schema,
action, and actor.
This function is most useful in conjunction with scope/4, which
builds an Ecto query that filters to only those resources the actor
is authorized for. If you run the resulting query and receive [], it
is not possible to determine whether the result is empty because the
actor wasn't authorized for any resources or because of other
restrictions on the query.
For example, you might use the following pattern to load all the
resources a user is allowed to read that were inserted in the last day:
query = from(r in MyResource, where: r.inserted_at > from_now(-1, "day"))

if any_authorized?(query, :read, user) do
  {:ok, scope(query, :read, user) |> Repo.all()}
else
  {:error, :not_authorized}
end
This would result in {:ok, results} if the user is authorized to
read any resources, even if the result set is empty, and would result
in {:error, :not_authorized} if the user isn't authorized to read
the resources at all.

  
  examples

  
  Examples


iex> MyPolicy.any_authorized?(MyResource, :read, actor)
true

iex> MyPolicy.any_authorized?(MyResource, :delete, actor)
false

  



    

  
    
      
      Link to this function
    
    authorize(resource, action, policy, opts \\ [])


      
       
       View Source
     


  


  

      

          @spec authorize(Ecto.Schema.t(), Janus.action(), Janus.Policy.t(), keyword()) ::
  {:ok, Ecto.Schema.t()} | {:error, :not_authorized}


      


Authorizes a loaded resource.
Expects to receive a struct, an action, and an actor or policy.
Returns {:ok, resource} if authorized, otherwise {:error, :not_authorized}.

  
  options

  
  Options


	:load_associations - Whether to load associations required by
policy authorization rules, defaults to false unless configured
on your policy module

	:repo - Ecto repository to use when loading required
associations if :load_associations is set to true, defaults to
nil unless configured on your policy module



  
  examples

  
  Examples


iex> MyPolicy.authorize(%MyResource{}, :read, actor) # accepts an actor
{:ok, %MyResource{}}

iex> MyPolicy.authorize(%MyResource{}, :read, policy) # or a policy
{:ok, %MyResource{}}

iex> MyPolicy.authorize(%MyResource{}, :delete, actor)
{:error, :not_authorized}

  



    

  
    
      
      Link to this function
    
    scope(query_or_schema, action, policy, opts \\ [])


      
       
       View Source
     


  


  

      

          @spec scope(filterable(), Janus.action(), Janus.Policy.t(), keyword()) ::
  Ecto.Query.t()


      


Create an %Ecto.Query{} that results in only authorized records.
Like the Ecto.Query API, this function can accept a schema as the
first argument or a query, in which case it will compose with that
query. If a query is passed, the appropriate schema will be derived
from that query's source.
scope(MyResource, :read, user)

query = from(r in MyResource, where: r.inserted_at > from_ago(1, "day"))
scope(query, :read, user)
If the query specifies the source as a string, we cannot derive the
schema. For example, this will not work:
# Raises an ArgumentError
query = from(r in "my_resources", where: r.inserted_at > from_ago(1, "day"))
scope(query, :read, user)

  
  options

  
  Options


	:preload_authorized - Similar to Ecto.Query.preload/3, but
only preloads those associated records that are authorized. Note
that this requires Ecto v3.9.4 or later and a database that supports
lateral joins. See "Preloading authorized associations" for more
information.


  
  preloading-authorized-associations

  
  Preloading authorized associations


The :preload_authorized option can be used to preload associated
records, but only those that are authorized for the given actor. An
additional query can be specified for each preloaded association that
will be run as if scoped to its parent row.
This can simplify certain queries dramatically. For instance, imagine
a user search interface that lists users along with their most recent
comment. Naughty comments can be hidden by moderators, but those
hidden comments should still be visible if a moderator is searching.
Here's how that might be accomplished:
iex> last_comment = from(Comment, order_by: [desc: :inserted_at], limit: 1)

iex> User
...> |> search(search_params)
...> |> MyPolicy.scope(:read, current_user,
...>   preload_authorized: [comments: last_comment]
...> )
...> |> Repo.all()
[%User{comments: [%Comment{}]}, %User{comments: [%Comment{}]}, ...]
Some things to note about this example:
	The last_comment query runs as if scoped to each user's
comments. This means that the :limit applies to each user's
comments, not the entire set of comments.

	The comment will be the last inserted comment that is authorized
to be read by the current_user. Moderators may be able to see
hidden comments, while normal users may not.


It is also possible to nest authorized preloads. For instance, you
could preload comments and their associated post.
MyPolicy.scope(User, :read, current_user,
  preload_authorized: [comments: :post]
)
This would load all comments. You could incorporate the last_comment
query above by specifying it as the first element of a tuple, followed
by the list of inner preloads:
MyPolicy.scope(User, :read, current_user,
  preload_authorized: [comments: {last_comment, [:post]}]
)
This would load only the latest comment as well as its associated post
(assuming it too is authorized to be read by current_user).

  
  examples

  
  Examples


iex> MyPolicy.scope(MyResource, :read, actor)
%Ecto.Query{}

iex> MyPolicy.scope(MyResource, :read, actor) |> Repo.all()
[%MyResource{}, ...]

iex> MyResource
...> |> MyPolicy.scope(:read, actor)
...> |> order_by(inserted_at: :desc)
...> |> limit(1)
...> |> Repo.one()
%MyResource{}

iex> MyResource
...> |> MyPolicy.scope(:read, actor,
...>   preload_authorized: :other
...> )
...> |> Repo.all()
[%MyResource{other: %OtherResource{}}, ...]

Define composable authorization policies for actors in your system.
A policy is a data structure created for an actor in your system that
defines the schemas that actor can access, the actions they can take,
and any restrictions to the set of resources that can be accessed.
These policies are generally created implicitly for actors passed to
functions defined by Janus.Authorization, but they can also be
created with build_policy/2.
Creating a policy modules
While you can create a policy module with use Janus.Policy, you will
usually invoke use Janus and implement build_policy/2:
defmodule MyApp.Policy do
  use Janus

  @impl true
  def build_policy(policy, _actor) do
    policy
  end
end
An implementation for build_policy/1 is injected into the policy
module.
Policy modules can now be used to generate policy structs explicitly
(though they will usually be created implicitly when calling functions
defined by Janus.Authorization).
iex> policy = MyApp.Policy.build_policy(:my_user)
%Janus.Policy{actor: :my_user, rules: %{...}}

iex> MyApp.SecondaryPolicy.build_policy(policy)
%Janus.Policy{actor: :my_user, rules: %{...}}
Permissions with allow and deny
Permissions are primarily defined using allow/4 and deny/4, which
allows or denies an action on a resource if a set of conditions match.
Both functions take the same arguments and options. When permissions
are being checked, multiple allow rules combine using logical-or,
with deny rules overriding allow.
For example, the following policy would allow a moderator to edit
their own comments and any comments flagged for review, but not those
made by an admin.
def build_policy(policy, %User{role: :moderator} = user) do
  policy
  |> allow(Comment, :edit, where: [user: [id: user.id]])
  |> allow(Comment, :edit, where: [flagged_for_review: true])
  |> deny(Comment, :edit, where: [user: [role: :admin]])
end
While set of keyword options passed to allow and deny are
reminiscent of keyword-based Ecto queries, but since they are
functions and not macros, there is no need to use the ^value syntax
used in Ecto. For example, the following would result in an error:
allow(policy, Comment, :edit, where: [user: [id: ^user.id]])
:where and :where_not conditions
These conditions match if the associated fields are equal to each
other. For instance, the moderation example above could also be
written as:
def build_policy(policy, %User{role: :moderator} = user) do
  policy
  |> allow(Comment, :edit, where: [user_id: user.id])
  |> allow(Comment, :edit,
    where: [flagged_for_review: true],
    where_not: [user: [role: :admin]]
  )
end
Multiple conditions within the same allow/deny are combined with a
logical-and, so this might be translated to English as "allow
moderators to edit comments they made or to edit comments flagged for
review that were not made by an admin".
:or_where conditions
You can also use :or_where to combine with all previous conditions.
For instance, the two examples above could also be written as:
def build_policy(policy, %User{role: :moderator} = user) do
  policy
  |> allow(Comment, :edit,
    where: [flagged_for_review: true],
    where_not: [user: [role: :admin]],
    or_where: [user_id: user.id]
  )
end
An :or_where condition applies to all clauses before it. Using some
pseudocode for demonstration, the above would read:
# (flagged_for_review AND NOT user.role == :admin) OR user_id == user.id
These clauses could be reordered to have a different meaning:
policy
|> allow(Comment, :edit,
  where: [flagged_for_review: true],
  or_where: [user_id: user.id],
  where_not: [user: [role: :admin]]
)

# (flagged_for_review OR user_id == user.id) AND NOT user.role == :admin
Attribute checks with functions
When equality is not a sufficient check for an attribute, a function
can be supplied.
For instance, a published_at field might be used to schedule posts.
Users may only have permission to read posts where published_at is
in the past, but we can only check for equality using the basic
keyword syntax presented above. In these cases, you can defer this
check using an arity-3 function:
def build_policy(policy, _actor) do
  policy
  |> allow(Comment, :read, where: [published_at: &in_the_past?/3])
end

def in_the_past?(:boolean, record, :published_at) do
  if value = Map.get(record, :published_at) do
    DateTime.compare(DateTime.utc_now(), value) == :gt
  end
end

def in_the_past?(:dynamic, binding, :published_at) do
  now = DateTime.utc_now()
  Ecto.Query.dynamic(^now > as(^binding).published_at)
end
As seen in the example above, functions must define at least two
clauses based on their first argument, :boolean or :dynamic, so
that they can handle both operations on a single record and operations
that should compose with an Ecto query.
Hooks
Functions can be registered as hooks that run prior to authorization
calls. See attach_hook/4 for more information.

        build_policy(arg1)

      


        Builds an authorization policy, delegating to build_policy/2.



    


    
      
        build_policy(t, actor)

      


        Builds an authorization policy containing rules for the given actor.



    





  
    Functions
  


    
      
        allow(policy, schema, action, opts \\ [])

      


        Allows an action on the schema if matched by conditions.



    


    
      
        allows(action)

      


        Specifies that a condition should match if another action is allowed.



    


    
      
        attach_hook(policy, name, schema \\ :all, fun)

      


        Attach a hook to the policy.



    


    
      
        attach_new_hook(policy, name, schema \\ :all, fun)

      


        Attach a new hook to the policy.



    


    
      
        deny(policy, schema, action, opts \\ [])

      


        Denies an action on the schema if matched by conditions.



    


    
      
        detach_hook(policy, name, schema \\ :all)

      


        Detach a hook from the policy.

      Link to this type
    
    hook()


      
       
       View Source
     


  


  

      

          @type hook() ::
  (Ecto.Schema.t() | Janus.Authorization.filterable(), Janus.action(), t() ->
     {:cont, Ecto.Schema.t() | Janus.Authorization.filterable()} | :halt)


      



  



  
    
      
      Link to this type
    
    t()


      
       
       View Source
     


  


  

      

          @type t() :: %Janus.Policy{
  actor: Janus.actor(),
  config: map(),
  hooks: %{optional(Janus.schema_module() | :all) => keyword(hook())},
  rules: %{
    required({Janus.schema_module(), Janus.action()}) => Janus.Policy.Rule.t()
  }
}

      Link to this callback
    
    build_policy(arg1)

          @callback build_policy(t() | Janus.actor()) :: t()


      


Builds an authorization policy, delegating to build_policy/2.
If given a policy, calls build_policy/2 with the policy and the
actor associated with the policy. If given an actor, creates an empty
policy associated with that actor and passes it to build_policy/2.
An implementation for this callback is injected into modules invoking
either use Janus or use Janus.Policy.

  



  
    
      
      Link to this callback
    
    build_policy(t, actor)

          @callback build_policy(t(), Janus.actor()) :: t()


      


Builds an authorization policy containing rules for the given actor.
See Janus.Policy for API documentation on building policies.

      Link to this function
    
    allow(policy, schema, action, opts \\ [])

          @spec allow(t(), Janus.schema_module(), Janus.action() | [Janus.action()], keyword()) ::
  t()


      


Allows an action on the schema if matched by conditions.
See "Permissions with allow and deny" for a description of conditions.

  
  examples

  
  Examples


policy
|> allow(FirstResource, :read)
|> allow(SecondResource, :create, where: [creator: [id: user.id]])

  



  
    
      
      Link to this function
    
    allows(action)

Specifies that a condition should match if another action is allowed.
If used as the value for an association, the condition will match if
the action is allowed for the association.

  
  examples

  
  Examples


Allow users to edit any posts they can delete.
policy
|> allow(Post, :edit, where: allows(:delete))
|> allow(Post, :delete, where: [user_id: user.id])
Don't allow users to edit posts they can't read.
policy
|> allow(Post, :read, where: [archived: false])
|> allow(Post, :edit, where: [user_id: user.id])
|> deny(Post, :edit, where_not: allows(:read))

  
  example-with-associations

  
  Example with associations


Let's say we have some posts with comments. Posts are visible unless
they are archived, and all comments of visible posts are also visible.
To start, we can duplicate the condition:
policy
|> allow(Post, :read, where: [archived: false])
|> allow(Comment, :read, where: [post: [archived: false]])
If we add additional clauses to the condition for posts, however, we
will have to duplicate them for comments. We can use allows instead:
policy
|> allow(Post, :read, where: [archived: false])
|> allow(Comment, :read, where: [post: allows(:read)])
Now let's say we add a feature that allows for draft posts, which
should not be visible unless a published_at is set. We can modify
only the condition for Post and that change will propogate to
comments.
policy
|> allow(Post, :read, where: [archived: false], where_not: [published_at: nil])
|> allow(Comment, :read, where: [post: allows(:read)])

  



    

  
    
      
      Link to this function
    
    attach_hook(policy, name, schema \\ :all, fun)

          @spec attach_hook(t(), atom(), :all | Janus.schema_module(), hook()) :: t()


      


Attach a hook to the policy.
Expects the following arguments:
	policy - the %Janus.Policy{} struct to attach to
	name - an atom identifying the hook
	schema (default :all) - an Ecto schema module identifying the
resource or query source that the hook should be applied to
	fun - the hook function, see "Hooks" below

If the given name is already present, an error will be raised. If
you wish to replace a hook, you can use detach_hook/3 before
re-attaching the hook. If you only wish to add a hook if it is hasn't
already been added, use attach_new_hook/4 instead.
Hooks will be run in the order that they are attached.

  
  hooks

  
  Hooks


Hooks are anonymous or captured functions that accept three arguments:
	operation - one of :authorize or :scope
	object - either a struct (for :authorize) or a queryable (for
:scope) that is being authorized
	action - the action being authorized

Hooks must return one of the following:
	{:cont, object} - additional hooks and authorization continue
	:halt - halt authorization, running no additional hooks and
returning {:error, :not_authorized} for an :authorize
operation and an empty query for a :scope operation


  
  examples

  
  Examples


When writing hooks, you must ensure that all possible arguments are
handled. This can be done using a "catch-all" clause. For example:
policy
|> attach_hook(:preload_user, fn
  :authorize, %Post{} = resource, _action ->
    {:cont, Repo.preload(resource, :user)}

  _operation, object, _action ->
    {:cont, object}
end)

policy
|> attach_hook(:preload_user, Post, fn
  :authorize, resource, _action ->
    {:cont, Repo.preload(resource, :user)}

  _operation, object, _action ->
    {:cont, object}
end)
Hooks can also be captured functions. For example:
policy
|> attach_hook(:preload_user, Post, &preload_user/3)

# elsewhere in your module

defp preload_user(:authorize, resource, _action) do
  {:cont, Repo.preload(resource, :user)}
end

defp preload_user(:scope, query, _action) do
  {:cont, from(query, preload: :user)}
end
If :halt is returned from a hook, no further hooks will be run and
nothing will be authorized. This could be used to perform a check on
banned users, for example:
@impl true
def build_policy(policy, user) do
  policy
  |> attach_hook(:ensure_unbanned, fn _op, object, _action ->
    if Accounts.banned?(user.id) do
      :halt
    else
      {:cont, object}
    end
  end)
end
This may be required if policies are being cached, since the hook runs
every time the authorization call happens, instead of only once when
the policy is built.

  



    

  
    
      
      Link to this function
    
    attach_new_hook(policy, name, schema \\ :all, fun)

          @spec attach_new_hook(t(), atom(), :all | Janus.schema_module(), hook()) :: t()


      


Attach a new hook to the policy.
Like attach_hook/4, except it only attaches the hook if the name
isn't present for the given schema.

    deny(policy, schema, action, opts \\ [])

          @spec deny(t(), Janus.schema_module(), Janus.action() | [Janus.action()], keyword()) ::
  t()


      


Denies an action on the schema if matched by conditions.
See "Permissions with allow and deny" for a description of conditions.

  
  examples

  
  Examples


policy
|> allow(FirstResource, :read)
|> deny(FirstResource, :read, where: [scope: :private])

  



    

  
    
      
      Link to this function
    
    detach_hook(policy, name, schema \\ :all)

          @spec detach_hook(t(), atom(), :all | Janus.schema_module()) :: t()


      


Detach a hook from the policy.

Defines a rule for an individual schema and action.

      Link to this type
    
    t()

          @type t() :: %Janus.Policy.Rule{
  action: Janus.action(),
  allow: [keyword() | boolean()],
  deny: [keyword() | boolean()],
  schema: Janus.schema_module()
}


      



  


        

      



  

    
mix janus.gen.policy 
    



      
Creates a basic policy module.
$ mix janus.gen.policy [--module MODULE] [--path PATH]

Creates the following file:
	lib/app_name/policy.ex - A Janus policy module containing
additional authorization helpers (see Generated Policy Modules
for more info)

Options
	--module - The name of the generated module, defaults to
AppName.Policy
	--app - The name of the application namespace, defaults to your
application name camelized, e.g. AppName
	--path - The path (including filename) for the generated module,
defaults to lib/app_name/policy.ex

